```
>
```

# Tangent Planes and Linear Approximations

```
> restart:with(plots):with(plottools):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

## Local Linearity (Zooming In) on a Graph of One Variable

We begin by focusing on the function $y = 4 - x^2$ and the tangent line $y = 5 - 2x$ to that curve at the point $(1, 3)$ .

```
> y:=4-x^2;
```

$$y := 4 - x^2$$

```
> p0:=plot(y,x=0..2):
  p1:=disk([1,3],.1,color=blue):
  p2:=plot(5-2*x,x=0..2,color=green):
  display(p0,p1,p2,scaling=constrained);
```




Now we zoom in on the point (1, 3) by a factor of 10.

```
> p0:=plot(y,x=.9..1.1):
  p1:=disk([1,3],.01,color=blue):
  p2:=plot(5-2*x,x=.9..1.1,color=green):
```

```
display(p0,p1,p2,scaling=constrained);
```




We zoom in on the point (1, 3) by another factor of 10 (overall, a factor of 100).

```
> p0:=plot(y,x=.99..1.01):
  p1:=disk([1,3],.001,color=blue):
  p2:=plot(5-2*x,x=.99..1.01,color=green):
  display(p0,p1,p2,scaling=constrained);
```




With this microscopic view, the original graph now looks pretty much like a straight line, in fact the tangent line. This phenomenon happens at any point on a function of one variable where the function is differentiable.

```
>
```

## Local Linearity (Zooming In) on a Graph of Two Variables

Let's now see what happens when we zoom in on a point on the surface of a two variable function where both partial derivatives exist and are continuous. We first need to reset the variable *y*.

```
> y:='y';
```

$$y := y$$

We will focus on the point (1, 1, 2) on the surface of the function $z = 5 - 2x^2 - y^4$.

```
> z:=5-2*x^2-y^4;
```

$$z := 5 - 2x^2 - y^4$$

```
> p0:=plot3d(z,x=0..2,y=0..2,shading=zhue,style=patchcontour):
  p1:=sphere([1,1,2],.1):
  display(p0,p1,axes=frame);
```




Click on the graph and change it's orientation to [10, 120]. Then we zoom in by a factor of 10.

```
> p0:=plot3d(z,x=.9..1.1,y=.9..1.1,shading=zhue,style=
  patchcontour):
  p1:=sphere([1,1,2],.01):
  display(p0,p1,axes=frame);
```




Again click on the graph and change its orientation to [10, 120]. We zoom in by another factor of 10.

```
> p0:=plot3d(z,x=.99..1.01,y=.99..1.01,shading=zhue,style=
  patchcontour):
  p1:=sphere([1,1,2],.001):
  display(p0,p1,axes=frame);
```




Once more, click on the graph and change its orientation to [10, 120]. It appears that we are looking at a plane.

Let's also approach the issue by way of contour diagrams corresponding to the three graphs above. From the first graph we get the folowing contours.

```
> p0:=contourplot(z,x=0..2,y=0..2):
  p1:=disk([1,1],.01):
  display(p0,p1,scaling=constrained);
```




We zoom in by a factor of 10.

```
> p0:=contourplot(z,x=.9..1.1,y=.9..1.1):
  p1:=disk([1,1],.001):
  display(p0,p1,scaling=constrained);
```




We zoom in by another factor of 10.

```
> p0:=contourplot(z,x=.99..1.01,y=.99..1.01):
  p1:=disk([1,1],.0001):
  display(p0,p1,scaling=constrained);
```




From this perspective, all of our contours appear to be lines. This would indicate that the corresponding surface would be a plane. We suspect that it is the tangent plane to the surface at the point (1, 1, 2). Using the zoom factor of 100 and the orientation of [10, 120], we view the surface and its tangent plane near (1, 1, 2).

```
> zz:=2-4*(x-1)-4*(y-1);
```

$$zz := 10 - 4\,x - 4\,y$$

```
> p0:=plot3d(z,x=.99..1.01,y=.99..1.01,shading=zhue,style=
  patchcontour):
  p1:=sphere([1,1,2],.001):
  p2:=plot3d(zz,x=.99..1.01,y=.99..1.01,shading=zhue,style=
  patchcontour):
  display(p0,p1,p2,axes=frame,orientation=[10,120]);
```




Visually, we cannot distinguish between the two graphs. We look at their contour plots, using red for the function and blue for the tangent plane.

```
> p0:=contourplot(z,x=.99..1.01,y=.99..1.01,color=red):
  p1:=disk([1,1],.0001):
  p2:=contourplot(zz,x=.99..1.01,y=.99..1.01,color=blue):
  display(p0,p1,p2,scaling=constrained);
```




We see their contour plots are basically indistinguishable also. We conclude that near a point of differentiability, the graph is very closely approximated by its tangent plane at the point.

>